

1990

## Correction

R. Benson

K. Benson

Follow this and additional works at: <https://digitalcommons.usf.edu/nabb>

---

### Recommended Citation

Benson, R. and Benson, K. (1990) "Correction," *North American Bird Bander*. Vol. 15 : Iss. 1 , Article 7.  
Available at: <https://digitalcommons.usf.edu/nabb/vol15/iss1/7>

This Contents is brought to you for free and open access by the Searchable Ornithological Research Archive at Digital Commons @ University of South Florida. It has been accepted for inclusion in North American Bird Bander by an authorized editor of Digital Commons @ University of South Florida. For more information, please contact [digitalcommons@usf.edu](mailto:digitalcommons@usf.edu).

## Correction .....

Following is the corrected computer program for the Tandy PC-4 only which appeared in The Pocket Computer: A New Tool For Identifying Eastern *Empidonax* Flycatchers in the Hand by R. Benson and K. Benson (Dept. of Engineering Tech., Texas A&M University, College Station, TX 77843), *NABB*, Vol. XIV, No. 3 (July - September 1989), p. 82.

```
5 INPUT "WC = ",A
10 INPUT "PE = ",B
15 INPUT "BL = ",C
20 INPUT "BW = ",D
25 INPUT "TL = ",E
30 INPUT "P6 = ",F
35 G=A*4.452-B*.99+C*31.656+D*38.102+E*6.698+F*35.836-543.43
40 H=A*4.17+B*1.938+C*47.279+D*52.325+E*6.82+F*57.742-818.671
45 I=A*4.715+B*.635+C*36.748+D*44.708+E*5.489+F*40.044-593.403
50 J=A*4.899-B*.44+C*42.405+D*45.929+E*7.032+F*51.597-750.081
55 L=G
60 IF H>L THEN L=H
65 IF I>L THEN L=I
70 IF J>L THEN L=J
75 IF G=L THEN PRINT "LEAST"
80 IF H=L THEN PRINT "ACADIAN"
85 IF I=L THEN PRINT "YEL BELLD"
90 IF J=L THEN PRINT "TRAILL"
95 P=(1/( EXP(G-L)+ EXP(H-L)+ EXP(I-L)+ EXP(J-L) ) *100)
100 PRINT "PRB = ";
110 $= STR$(P)
125 PRINT MID$(1,5)
130 GOTO 5
```

In addition to the above program, the authors are submitting the following IBM or 100% compatible computer program written in Microsoft Basic.

```
100 CLS
110 LOCATE 2,24 : INPUT "Wing chord" "A"
120 LOCATE 3,24 : INPUT "Primary extension" "B"
130 LOCATE 4,24 : INPUT "Bill length" "C"
140 LOCATE 5,24 : INPUT "Bill width" "D"
150 LOCATE 6,24 : INPUT "Tail length" "E"
160 LOCATE 7,24 : INPUT "P6 emargination" "F"
165
170 G=A*4.452-B*.99+C*31.656+D*38.102+E*6.698+F*35.836-543.43
180 H=A*4.17+B*1.938+C*47.279+D*52.325+E*6.82+F*57.742-818.671
190 I=A*4.715+B*.635+C*36.748+D*44.708+E*5.489+F*40.044-593.403
200 J=A*4.899-B*.44+C*42.405+D*45.929+E*7.032+F*51.597-750.081
210 L=G
220 IF H>L THEN L=H
230 IF I>L THEN L=I
240 IF J>L THEN L=J
250 IF G=L THEN SPECIES$ = "LEAST FLYCATCHER"
260 IF H=L THEN SPECIES$ = "ACADIAN FLYCATCHER"
270 IF I=L THEN SPECIES$ = "YELLOW-BELLIED FLYCATCHER"
280 IF J=L THEN SPECIES$ = "TRAILL'S FLYCATCHER"
290 PROB=(1/( EXP(G-L)+ EXP(H-L)+ EXP(I-L)+ EXP(J-L) ) *100)
300 LOCATE 14,27 : PRINT SPECIES$
310 LOCATE 17,2
320 PRINT "Probability of correct classification based on these data is ";
330 PRINT USING "###.###",PROB; : PRINT " %"
340 LOCATE 22,10 : PRINT "Press <ESC> to enter new data, any other key to end";
350 TEST$ = INKEY$ : IF TEST$ = "" THEN GOTO 350
360 IF TEST$ = CHR$(27) THEN GOTO 100 ELSE SYSTEM
```

Note: This program contains no error checking. It will force a classification even with input values lying completely out of normal range. Be sure to enter data correctly.

The authors have a user-friendly version with error checking of the above program which they will send free of charge to any bander or researcher who will provide them with a formatted disk (either 5.25" or 3.5"). Any comments as to the use of these computer programs would be appreciated by the authors.